Mathematisches Institut
der Universität München
Prof. Dr. Helmut Schwichtenberg
Dr. Christian Urban

Wintersemester 2004/2005
Blatt 4

Übungen zum Schemekurs

Aufgabe 13. Man betrachte den folgenden "Operator der primitiven Rekursion"

```
(define (prim-rec a step)
  (letrec ((f (lambda (n)
                (if (zero? n)
                    a
                    (step n (f (- n 1)))))))
    f))
```

(1) Programmieren Sie eine iterative Fassung dieses Operators.
(2) Man betrachte die primitiv rekursive Funktion $f\colon \mathbb{N} \to \mathbb{N}$ mit

$$\begin{aligned} f(0) &= 1 \\ f(n) &= f(n-1) \cdot (f(n-1) + 1) \quad (n > 0) \end{aligned}$$

Definieren Sie explizit aus dem Operator der primitiven Rekursion eine Prozedur zur Berechnung von f.

Lösung.

```
(define (prim-rec a step)
  (lambda (n) (prim-rec3 a step n)))

(define (prim-rec3 a step n)
  (prim-rec-iter step n 0 a))

(define (prim-rec-iter step n count acc)
  (if (= count n)
      acc
      (prim-rec-iter step n (+ count 1) (step (+ count 1) acc))))

(define f (prim-rec 1 (lambda (n b) (* b (+ b 1)))))
```

Aufgabe 14. Ein Polynom $a_n X^n + \cdots + a_1 X + a_0$ mit $a_n \neq 0$ sei als Liste $(a_0 \ \ldots \ a_n)$ dargestellt. Zu schreiben sind die Funktionen:

(1) `poly+`, die die Summe zweier Polynome berechnen soll
(2) `poly*skal`, die das Produkt eines Polynomes mit einer Zahl berechnen soll, und
(3) `poly*`, die das Produkt zweier Polynome berechnen soll.

Lösung.

```
(define (poly+ a b)
  (cond ((= (length a) (length b)) (map + a b))
        ((> (length a) (length b)) (map + a (make-same-length a b)))
        ((< (length a) (length b)) (map + (make-same-length b a)) b)))

(define (make-same-length a b)
  (if (= (length a) (length b))
      b
      (make-same-length a (append b (list 0)))))

(define (poly*skal p skal)
  (if (= skal 0)
      '()
      (map (lambda (x) (* x skal)) p)))

(define (poly* p q)  ; p und q Polynome
  (if (null? p)
      '()       ; 0*q=0
      (poly+ (poly*skal q (car p)) (cons 0 (poly* (cdr p) q)))))
                ; sonst p = p0 + X*p1, und p*q = p0*q + X*(p1*q)
```

Aufgabe 15.* Man schreibe eine Funktion `polydivide`, die zu ihren Argumenten p und q (Polynome) Quotient a und Rest r berechnet (d.h. $p = a \cdot q + r$). Es empfiehlt sich, hier die Polynome "von hinten", d.h. beginnend mit dem Leitkoeffizienten, zu bearbeiten. Deswegen werden die Polynome erst einmal umgedreht. Außerdem ist es für die Rechnung praktischer, wenn man die Polynome so normalisiert, daß q normiert ist, d.h. Leitkoeffizient 1 hat. Der Quotient ändert sich dadurch nicht; der Rest muß am Ende wieder mit dem Leitkoeffizienten von q multipliziert werden.

Lösung.

```
(define (polydivide p q)  ; p und q Polynome
  (if (null? q)
      (begin (display #\newline)
             (display "Division durch Null!")
             '(() ()))
          ; Wenn q=0 ist, Warnung ausgeben, zwei Nullpolynome als Wert
      (let* ((pr (reverse p)) ; pr=p umgedreht
             (qr (reverse q)) ; qr=q umgedreht
             (lkfq (car qr))  ; lkfq = Leitkoeffizient von q
             (l (polydivide1 (poly*skal pr (/ lkfq))
                             (poly*skal qr (/ lkfq)))))
                ; l wird an die Liste (a' r') gebunden
        (list (reverse (car l)) (poly*skal (reverse (cadr l)) lkfq)))))
           ; jetzt wird das Ergebnis (a r) zusammengebaut: a entsteht aus
           ; a' durch Umdrehen, r aus r' durch Umdrehen und Multiplikation
           ; mit q0.


(define (polydivide1 p q)  ; p und q umgedrehte Polynome, q normiert
  (cond ((null? p) (list '() '())) ; p=0, dann a'=r'=0
        ((zero? (car p)) (polydivide1 (cdr p) q))
           ; fuehrende Null beseitigen
        ((< (length p) (length q)) (list '() p))
                    ; deg(p) < deg(q), dann a'=0, r'=p
        (else (let ((l (polydivide1
                        (polydivide2 (car p) (cdr p) (cdr q)) q)))
          ; polydivide2 subtrahiert das (car p)-fache von q "von vorn"
          ; von p, l ist die Liste (a'' r') aus dem Quotienten dieser
          ; Differenz mit q und dem Rest (der bereits der Rest
          ; von p mod q ist)
                (list (cons (car p) (car l)) (cadr l))))))
                   ; Es ist a'=(Leitkoeff. von p)*X**n + a'', n passend

(define (polydivide2 a pl ql)
          ; a Zahl, pl, ql Listen, length(pl)>=length(ql)
  (if (null? ql)
      pl     ; ql ist aufgebraucht, pl bleibt uebrig
      (cons (- (car pl) (* a (car ql)))
            (polydivide2 a (cdr pl) (cdr ql)))))
             ; sonst vorne Subtrahieren
             ; und an verarbeiteten Rest anhaengen
```

Abgabe. Freitag, den 8. Oktober 2004, vor der Vorlesung oder (besser) per Email an `urban@mathematik.uni-muenchen.de`